Вот есть дополнительный реквизит справочника "ДоговорыКонтрагентов" с названием процент по договору

И хочу выбрать остатки по договорам займа вместе со значением этого реквизита.

Вот запрос

```1c
    Запрос.Текст="ВЫБРАТЬ
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт) КАК НачОстаток,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотДт) КАК ПРИХОД,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотКт) КАК РАСХОД,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт) КАК КонОстаток,
    | ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Контрагент,
    | ХозрасчетныйОстаткиИОбороты.Субконто2 КАК Договор,
    | МАКСИМУМ(ДоговорДополнительныеРеквизиты.Свойство) КАК Свойство,
    | МАКСИМУМ(ЕСТЬNULL(ДоговорДополнительныеРеквизиты.Значение, 0)) КАК ПроцентПоДоговору
    |ПОМЕСТИТЬ ВТ_ИТОГИ
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, Месяц, ДвиженияИГраницыПериода, Счет = ЗНАЧЕНИЕ(ПланСчетов.Хозрасчетный.ПредоставленныеЗаймы), &ВидыСубконто, Субконто1 = &Контрагент) КАК ХозрасчетныйОстаткиИОбороты
    | ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ДоговорыКонтрагентов.ДополнительныеРеквизиты КАК ДоговорДополнительныеРеквизиты
    | ПО (ХозрасчетныйОстаткиИОбороты.Субконто2 = ДоговорДополнительныеРеквизиты.Ссылка)
    |ГДЕ
    | ДоговорДополнительныеРеквизиты.Свойство = &Свойство
    |
    |СГРУППИРОВАТЬ ПО
    | ХозрасчетныйОстаткиИОбороты.Субконто1,
    | ХозрасчетныйОстаткиИОбороты.Субконто2";

    Запрос.УстановитьПараметр("НачалоМесяца",НачалоМесяца);
    Запрос.УстановитьПараметр("КонецМесяца",КонецМесяца);
    Запрос.УстановитьПараметр("Организация",Организация);
    Запрос.УстановитьПараметр("Контрагент",ВыбКонтрагент);

    ВидыСубконто = Новый Массив;
    ВидыСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Контрагенты);
    ВидыСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры);
    Запрос.УстановитьПараметр("ВидыСубконто", ВидыСубконто);
    Запрос.УстановитьПараметр("Свойство", ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Процент по договору"));
```

Но остатки выводятся только по договорам, где заполнен этот реквизит. А по тем, где не заполнен, не выводится.
---
Дело в том, что Вы накладываете условие на поле присоединяемой таблицы дополнительных реквизитов. В этом случае ЛЕВОЕ соединение становится ВНУТРЕННИМ. Поэтому то Вы и видите только те договоры, где заполнено доп. свойство.

Перенесите условие отбора по свойству из секции ГДЕ в условие связи:

```1c
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ДоговорыКонтрагентов.ДополнительныеРеквизиты КАК ДоговорДополнительныеРеквизиты
    ПО (ХозрасчетныйОстаткиИОбороты.Субконто2 = ДоговорДополнительныеРеквизиты.Ссылка)
    И ДоговорДополнительныеРеквизиты.Свойство = &Свойство
```
---
Дело в том, что Вы накладываете условие на поле присоединяемой таблицы дополнительных реквизитов. В этом случае ЛЕВОЕ соединение становится ВНУТРЕННИМ. Поэтому то Вы и видите только те договоры, где заполнено доп. свойство.

Перенесите условие отбора по свойству из секции ГДЕ в условие связи:

```1c
    ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ДоговорыКонтрагентов.ДополнительныеРеквизиты КАК ДоговорДополнительныеРеквизиты
    ПО (ХозрасчетныйОстаткиИОбороты.Субконто2 = ДоговорДополнительныеРеквизиты.Ссылка)
    И ДоговорДополнительныеРеквизиты.Свойство = &Свойство
```
---
если грубо, то так

```1c
    ....

    |ГДЕ
    | ДоговорДополнительныеРеквизиты.Свойство = &Свойство

    ОБЪЕДИНИТЬ ВСЕ
    .....
    |ГДЕ
    | ДоговорДополнительныеРеквизиты.Свойство <> &Свойство
```
---
Нет, это работать не будет. Во-первых, потому что в обоих случаях накладывается условие на поле присоединяемой таблицы, что превратит левое соединение во внутреннее. А во-вторых, второй запрос вместо процентов по займу выдаст все остальные значения дополнительных реквизитов.
---
Ну да. Я понял, что проще создать две виртуальные таблицы, а потом уже их соединять.

```1c
    МенВрТаб = Новый МенеджерВременныхТаблиц;

    Запрос= Новый Запрос;
    Запрос.МенеджерВременныхТаблиц = МенВрТаб;

    Запрос.Текст="ВЫБРАТЬ
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт) КАК НачОстаток,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотДт) КАК ПРИХОД,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаОборотКт) КАК РАСХОД,
    | СУММА(ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт) КАК КонОстаток,
    | ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Контрагент,
    | ХозрасчетныйОстаткиИОбороты.Субконто2 КАК Договор
    |ПОМЕСТИТЬ ВТ_ИТОГИ
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачалоМесяца, &КонецМесяца, Месяц, ДвиженияИГраницыПериода, Счет = ЗНАЧЕНИЕ(ПланСчетов.Хозрасчетный.ПредоставленныеЗаймы), &ВидыСубконто, Субконто1 = &Контрагент) КАК ХозрасчетныйОстаткиИОбороты
    |
    |СГРУППИРОВАТЬ ПО
    | ХозрасчетныйОстаткиИОбороты.Субконто1,
    | ХозрасчетныйОстаткиИОбороты.Субконто2";

    Запрос.УстановитьПараметр("НачалоМесяца",НачалоМесяца);
    Запрос.УстановитьПараметр("КонецМесяца",КонецМесяца);
    Запрос.УстановитьПараметр("Организация",Организация);
    Запрос.УстановитьПараметр("Контрагент",ВыбКонтрагент);
    ВидыСубконто = Новый Массив;
    ВидыСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Контрагенты);
    ВидыСубконто.Добавить(ПланыВидовХарактеристик.ВидыСубконтоХозрасчетные.Договоры);
    Запрос.УстановитьПараметр("ВидыСубконто", ВидыСубконто);

    Запрос.Выполнить();

    //нужно наименование договора и процент
    Запрос = Новый Запрос;
    Запрос.МенеджерВременныхТаблиц = МенВрТаб;
    Запрос.Текст = "ВЫБРАТЬ
    | ДоговорДополнительныеРеквизиты.Ссылка.Наименование КАК НАИМЕНОВАНИЕ,
    | МАКСИМУМ(ДоговорДополнительныеРеквизиты.Свойство) КАК Свойство,
    | МАКСИМУМ(ДоговорДополнительныеРеквизиты.Значение) КАК ПроцентПоДоговору
    |ПОМЕСТИТЬ ВТ_Процент
    |ИЗ
    | Справочник.ДоговорыКонтрагентов.ДополнительныеРеквизиты КАК ДоговорДополнительныеРеквизиты
    |ГДЕ
    | ДоговорДополнительныеРеквизиты.Свойство = &Свойство
    | И ДоговорДополнительныеРеквизиты.Ссылка.Владелец = &Контрагент
    |
    |СГРУППИРОВАТЬ ПО
    | ДоговорДополнительныеРеквизиты.Ссылка.Наименование";

    Запрос.УстановитьПараметр("Свойство", ПланыВидовХарактеристик.ДополнительныеРеквизитыИСведения.НайтиПоНаименованию("Процент по договору"));
    Запрос.УстановитьПараметр("Контрагент",ВыбКонтрагент);
```